



Senter for
klimaforskning

Center for
International Climate
and Environmental
Research - Oslo

Working Paper 1999:1

**An efficient and
accurate carbon
cycle model for
use in simple
climate models**

Knut H. Alfsen and Terje Berntsen



Universitetet i Oslo

University of Oslo

ISSN: 0804-452X

CICERO Working Paper 1999:1

An efficient and accurate carbon cycle model for use in simple climate models

Knut H. Alfsen and Terje Berntsen

May 25 1999

CICERO

Center for International Climate and
Environmental Research – Oslo

P.O. Box 1129 Blindern

N-0317 Oslo, Norway

Phone: (+47) 22 85 87 50

Fax: (+47) 22 85 87 51

E-mail: admin@cicero.uio.no

Web: www.cicero.uio.no

Abstract

This publication describe a simple, but accurate, model for calculating CO₂ concentration levels in the atmosphere. The model is based on extracts from Joos, F., M. Bruno, R. Fink, U. Siegenthaler and T.F. Stocker (1996): An efficient and accurate representation of complex oceanic and biospheric models of antropogenic carbon uptake, *Tellus* 48B, 397-417.

Contents

1	INTRODUCTION	2
2	AIR-SEA INTERFACE MODELLING	2
2.1	Structure of the problem	5
3	BIOSPHERE DECAY RESPONSE FUNCTION	6
4	PROCEDURE OF CALCULATION OF ATMOSPHERIC CO ₂ CONCENTRATION	7
4.1	Calculating the biosphere response	9
	REFERENCES	9
	Appendix A: Fortran program	10

1 Introduction

The aim of this note is to document a simple model of the atmospheric CO₂ concentration based on exogenous input of anthropogenic emission of CO₂ and taking air-sea exchange and biospheric responses into account.

The approach described by Joos et al (1996) is based on the application of a mixed layer pulse response function. The advantage of using a mixed layer pulse response function instead of an atmosphere pulse response function (Siegentaler and Oeschger, 1978; Oeschger and Heimann, 1983; Meier-Reimer and Hasselmann, 1987; Sarmiento et al., 1992) is that it is then possible to represent the non-linear effects of seawater chemistry. As long as the CO₂ increase in the atmosphere is below approximately 50% of the pre-industrial level, the CO₂ system behaves in a linear way and it is possible to represent the effects of anthropogenic emissions on the atmospheric concentrations by an atmospheric pulse response function. For CO₂ perturbations beyond this level the non-linear effects of the seawater chemistry becomes important, and it thus becomes necessary to apply a mixed layer pulse response function to obtain accurate results.

The approach described in this paper includes changes in CO₂ uptake and release by terrestrial vegetation by CO₂ fertilization, but does not take into account possible feedback mechanisms of climate change on the carbon cycle. Possible feedbacks include changes in CO₂ solubility due to sea surface temperature (SST) changes and changes in vertical mixing by reduced deep-water formation in the North Atlantic. Joos et al. (1999a) have estimated the marine part of this to be of minor importance (approximately 4%) up to 2100, increasing to about 20% in 2500.

Since damage caused by climate change could imply high costs, an accurate representation of the carbon cycle is very important in models that are to be used to estimate the costs of climate and evaluate different mitigation and/or adaptation strategies (Joos et al., 1999b). The carbon cycle model described in this paper is included in a simple model for scenario studies of changes in global climate (Fuglestad and Berntsen, 1999).

2 Air-sea interface modelling

The following expressions form the basis for the modelling of air-sea exchange of CO₂.

- The content of inorganic carbon in the ocean's mixing layer:

$$\delta\Sigma CO_2(t) \equiv \Sigma CO_2(t) - \Sigma CO_2(t_0) = \frac{c}{h} \int_{t_0}^t f_{as}(t') r_s(t-t') dt' \quad (1)$$

where h is the height of the mixing layer (typically 75 m) and c is a conversion factor translating the air-sea flux from $[ppm \cdot m^{-2} yr^{-1}]$ to $[\mu mol \cdot m^{-2} yr^{-1}]$ and sea water from $[m^3]$ to $[kg]$. We will apply the value $c = 1.722 \cdot 10^{17} \frac{\mu mol \cdot m^3}{ppm \cdot kg}$. The variable $\Sigma CO_2(t)$ is the content of inorganic carbon in the mixing layer, f_{as} is the transfer (flux) of CO_2 from air to the sea and r_s is the pulse response function of the *mixing layer*.

- The pulse response function is given by two expressions:

$0 < t < 2$ year:

$$r_s(t) = 0.12935 + 0.21898e^{-\frac{t}{0.034569}} + 0.17003e^{-\frac{t}{0.26936}} \\ + 0.24071e^{-\frac{t}{0.96083}} + 0.24093e^{-\frac{t}{4.9792}}$$

Mixing layer response function for $t < 2$ yr

and for $t \geq 2$ year:

$$r_s(t) = 0.022936 + 0.24278e^{-\frac{t}{1.2679}} + 0.13963e^{-\frac{t}{5.2528}} + 0.089318e^{-\frac{t}{18.601}} \\ + 0.037820e^{-\frac{t}{68.736}} + 0.035549e^{-\frac{t}{232.3}}$$

Mixing layer response function for $t > 2$ years

- Carbon budget for the atmosphere:

The perturbation of the carbon concentration in the atmosphere from pre-industrial level (278 ppm) can be obtained from the conservation law equating the change in atmospheric concentration with the difference between (net) emissions and sea uptake:

$$\frac{d}{dt}\delta pCO_{2,a}(t) = e(t) - A_{oc}f_{as}(t) \quad (2)$$

where the unit is [$ppm \cdot yr^{-1}$]. Note that $1 ppm = 2.123 \times 10^{15} gC = 7.779 \times 10^{15} gCO_2$. The variable $pCO_{2,a}(t)$ is the partial pressure of CO_2 in the atmosphere, $e(t)$ is antropogenic emissions of CO_2 available for sea uptake (see below) and A_{oc} is the area of the ocean surface ($A_{oc} = 3.62 \cdot 10^{14} m^2$).

- Transfer rate air-sea:

The transfer rate between the atmosphere and the sea (mixing layer) depends on the difference in partial pressures:

$$f_{as}(t) = k_g [\delta pCO_{2,a}(t) - \delta pCO_{2,s}(t)] \quad (3)$$

where k_g is the gas exchange coefficient ($= \frac{1}{9.06 \cdot A_{oc}} yr^{-1} m^{-2}$).

- Partial CO_2 pressure in the mixing layer:

The partial CO_2 pressure in the mixing layer is a complex function of the carbon content in the layer and sea temperature T :

$$\delta pCO_{2,s}(t) = F(\delta \Sigma CO_2(t), T) \quad (4)$$

An usual value for T is $T = 18.2^\circ C$. Two functional forms of F are given. The first is a reasonable approximation to the true functional form when $0 < \delta pCO_{2,s}(t) < 200 ppm$ and $0^\circ C < T < 30^\circ C$:

$$\begin{aligned} F(x, T) &= \frac{(1.7561 - 31.618 \cdot 10^{-3}T + 444.4 \cdot 10^{-6}T^2) x}{1 - (4.096 \cdot 10^{-3} - 77.086 \cdot 10^{-6}T + 0.61 \cdot 10^{-6}T^2) x} \\ &= \frac{1.3279x}{1 - 2.8951 \times 10^{-3}x} \text{ when } T=18.2^\circ C \end{aligned}$$

First version of $F(x, T)$ with $T = 18.2$ °C

The second version covers the range $0 < \delta pCO_{2,s}(t) < 1320$ ppm, but is rather narrow with respect to temperature: $17.7^\circ C < T < 18.3^\circ C$:

$$\begin{aligned}
 F(x, T) = & \left(1.5568 - 13.993 \cdot 10^{-3}T\right) x + (7.4706 - .20207T) 10^{-3}x^2 \\
 & - (12.748 - 1.2015T) 10^{-6}x^3 + (244.91 - 12.639T) 10^{-9}x^4 \\
 & - (0.15468 - 0.015326T) 10^{-9}x^5
 \end{aligned}$$

which, with the chosen temperature T , can be simplified to

$$\begin{aligned}
 F(x, T = 18.2) = & 1.3021x + 3.7929 \times 10^{-3}x^2 + 9.1193 \times 10^{-6}x^3 \\
 & + 1.488 \times 10^{-8}x^4 + 1.2425 \times 10^{-10}x^5
 \end{aligned}$$

The second version of $F(x, T = 18.2)$

In our model we employ the second version of F with $T = 18.2^\circ C$.

2.1 Structure of the problem

The above equations 1-4 represent enough constraints to determine the four variables $\delta pCO_{2,a}$, $\delta pCO_{2,s}$, $\delta \Sigma CO_2$ and f_{as} . Employing eq. 3 to eliminate f_{as} results in equations in the three variable $\delta pCO_{2,a}$, $\delta pCO_{2,s}$ and $\delta \Sigma CO_2$.

For simplicity I will denote these variable x , y and z , respectively. We also introduce the variable $s(t) = x(t) - y(t)$ which is proportional to the air-sea carbon flux. The equations can then be expressed as follows (denoting the parameters k_g by k and A_{oc} by A)

$$x(t) - x(0) = x(t) = \int_0^t [e(t') - A k s(t')] dt', \quad (5)$$

$$z(t) = k \frac{c}{h} \int_0^t s(t') r_s(t - t') dt', \quad (6)$$

$$y(t) = F(z(t)), \quad (7)$$

$$s(t) = x(t) - y(t)$$

3 Biosphere decay response function

Emission of relevance for atmosphere-ocean couplings may be written

$$e(t) = e_{anthropogenic}(t) - f_{fer}(t) \quad (8)$$

where f_{fer} is the biospheric sink due to additional net primary production (δf_{npp}) and decay (δf_{decay}) from CO₂ fertilization:

$$f_{fer}(t) = \delta f_{npp} - \delta f_{decay} = \delta f_{npp}(t) - \int_{-\infty}^t \delta f_{npp}(t') r_b(t - t') dt' \quad (9)$$

The uptake of atmospheric CO₂ from increase in net primary production is modelled as

$$\delta f_{npp}(t) = 60 [GtC/yr] \beta \ln \frac{CO_{2,a}(t)}{278} \quad (1)$$

where $\beta = 0.287$ is the fertilization factor and the CO₂ concentration level is measured in ppm. The unit of $\delta f_{npp}(t)$ (and f_{fer}) is $GtC = 10^{15} gC = PgC$.

Net primary production response to CO₂ in the atmosphere

For $t > 0$ we use the following expression for the biotic decay response function

$$r_b(t) = 0.70211e^{-0.35t} + 13.414 \cdot 10^{-3} e^{-\frac{t}{20}} - 0.71846e^{-\frac{55t}{120}} + 2.9323 \cdot 10^{-3} e^{-\frac{t}{100}}$$

The form of the function is as follows:

Biosphere decay response function r_b

4 Procedure of calculation of atmospheric CO₂ concentration

In solving the above system of equations on a computer, a discretization is necessary. One possible discrete version of the equations 5-7 is (trapezoidal rule)

$$x_i - x_0 = \Delta t \left[\frac{1}{2} (e_1 - Ak(x_1 - y_1)) + \sum_{j=2}^{i-1} (e_j - Ak(x_j - y_j)) \right] \quad (11)$$

$$+\frac{1}{2}(e_i - Ak(x_i - y_i)),$$

$$y_i = F(z_i), \quad (12)$$

$$\begin{aligned} z_i &= \frac{ck}{h} \Delta t \times \left[\frac{1}{2} (x_1 - y_1) r_i + \sum_{j=2}^{i-1} [x_j - y_j] r_{i-j} + \frac{1}{2} (x_i - y_i) r_1 \right] \quad (13) \\ &= \frac{ck}{h} \Delta t \times \left[\sum_{j=1}^{i-1} [x_j - y_j] r_{i-j} - \frac{1}{2} (x_1 - y_1) r_i + \frac{1}{2} (x_i - y_i) r_1 \right]. \end{aligned}$$

Equation 11 can be solved with respect to x_i to yield

$$\begin{aligned} x_i &= c_1 \left[\sum_{j=2}^{i-1} \left(\frac{e_j}{Ak} - (x_j - y_j) \right) + \frac{1}{2} \left(\frac{e_1}{Ak} - (x_1 - y_1) + \frac{e_i}{Ak} + y_i \right) \right] \quad (14) \\ &= c_1 \left[\sum_{j=1}^{i-1} \left(\frac{e_j}{Ak} - (x_j - y_j) \right) - \frac{1}{2} \left(\frac{e_1}{Ak} - (x_1 - y_1) \right) + \frac{1}{2} \left(\frac{e_i}{Ak} + y_i \right) \right], \end{aligned}$$

where the coefficient c_1 is given by

$$c_1 = \frac{\Delta t Ak}{1 + \Delta t Ak/2}.$$

Using

$$s_i = x_i - y_i$$

we get

$$\begin{aligned} s_i &= c_1 \left[\sum_{j=1}^{i-1} \left(\frac{e_j}{Ak} - s_j \right) - \frac{1}{2} \left(\frac{e_1}{Ak} - s_1 \right) + \frac{1}{2} \left(\frac{e_i}{Ak} + y_i \right) \right] - y_i \quad (15) \\ &= c_1 \left[\sum_{j=1}^{i-1} \left(\frac{e_j}{Ak} - s_j \right) - \frac{1}{2} \left(\frac{e_1}{Ak} - s_1 \right) + \frac{1}{2} \frac{e_i}{Ak} - \frac{y_i}{\Delta t Ak} \right] \end{aligned}$$

$$z_i = \frac{ck}{h} \Delta t \times \left[\sum_{j=1}^{i-1} s_j r_{i-j} - \frac{1}{2} s_1 r_i + \frac{1}{2} s_i r_1 \right] \quad (16)$$

The recursive procedure is then:

1. from given y_i , calculate s_i from eq. (15),
2. calculate z_i from eq. (16)
3. finally calculate a new y_i from eq. (12).

4.1 Calculating the biosphere response

Similarly, using a trapezoidal scheme for integration of the biosphere equation, and approximating equation 10 by

$$\delta f_{npp,i} = 60\beta \ln \frac{CO_{2,a,i-1}}{278},$$

and further noting that $r_b(0) = 0$ and $\delta f_{npp}(0) = 0$, we get a discretized version of eq. 10 as follows

$$f_{fer,i} = \delta f_{npp,i} - \Delta t \sum_{j=1}^{i-1} \delta f_{npp,j} r_{b,i-j}. \quad (17)$$

References

- [1] Fuglestedt J.S. and T. Berntsen. A simple model for scenario studies of changes in global climate: Version 1.0. Working Paper 1999:2, CI-CERO,1999.
- [2] Joos F. M. Bruno, R. Fink, U. Siegenthaler, T. F. Stocker, C. Le Quere, and J.L. Sarmiento. An efficient and accurate representation of complex oceanic and biospheric models of anthropogenic carbon uptake. *Tellus*, 48B, 397-417, 1996.
- [3] Joos F., G.-K. Plattner, T.F. Stocker, O. Marchal and A. Schmittner. Global warming and marine carbon cycle feedbacks on the future atmospheric CO₂. *Science*, 284, 464-467, 1999a.
- [4] Joos F., G. Muller-Furstenberger, and G. Stephan. Correcting the carbon cycle representation: How important is it for the economics of climate change. *Environmental Modelling & Assessment*, in press,1999b.
- [5] Meier-Reimer E, and K. Hasselmann. Transport and storage of CO₂ in the ocean - an inorganic/ocean circulation carbon cycle model. *Climate Dynamics*, 2, 63-90.
- [6] Oeschger H. and M. Heimann. Uncertainties of predictions of future atmospheric CO₂ concentrations. *J. Geophys. Res.*, 88, 1258-1262, 1983.
- [7] Sarmiento J. L., J. Orr, and U Siegenthaler. A perturbation simulation of CO₂ uptake in an ocean general circulation model. *J. Geophys. Res.*, 97, 3621-3645, 1992.
- [8] Siegenthaler U. and H. Oeschger. Predicting future atmospheric carbon dioxide levels. *Science*, 199, 388-395, 1978.

5

A Fortran program

Below follows a Fortran-subroutine incorporating the above relations.

```
SUBROUTINE CO2EM2CONC(NY,NYE)
C
C — CO2 model based on Joos et al., Tellus, 48B, 397-417, 1996.
C — Parameter NY is number of years since NSTART(=1750)
C
  INCLUDE 'scm.cmn'
C
  REAL*8 SUMZ, SUMF, SS2, SS1
  REAL*8 K,CC1
C
C — Fertilization factor
  BETAF = 0.287
C — Area of the ocean (m2)
  A = 3.62E+14
C — Gas exchange coefficient (air <-> ocean) (yr-1*m-2)
  K = 1./(A*9.06)
C — TIMESTEP (YR)
  DT = 1./IDTM
C — Conversion factor ppm/kg -> umol*/m3
  C = 1.722E+17
C — Depth of mixed layer in the ocean (75 m)
  H = 75.
C
  CC1 = DT*A*K/(1.+DT*A*K/2.)
C
  DO IDT = 1,IDTM
  IT = (NY-1)*IDTM + IDT
C
C — Net emissions, including biogenic fertilization effects —————
```

```

C
EMFF = FLOAT(IDT)/IDTM
SUMF = 0.0
IF (IT .GT.1) THEN
DFNPP(IT) = 60.0*BETAF*LOG(XCO2(IT-1)/278.)
DO I=2,IT-1
ITX = IT-I
SUMF = SUMF + DFNPP(I)*RB(ITX)
END DO
ELSE
SUMF = 0.0
END IF
FFER = DFNPP(IT) - DT*SUMF
C
IF (NY .LT. 241 ) THEN
EMCO2(IT) = (DFCO2F(NY)+EMCO2F(NY))*(1.-EMFF) +
. (DFCO2F(NY+1)+EMCO2F(NY+1))*EMFF - FFER
ELSE
IF (NYE .LT. NYMAX ) THEN
EMCO2(IT) = EMISSION(1,NYE)*(1.-EMFF) +
. EMISSION(1,NYE+1)*EMFF - FFER
ELSE
EMCO2(IT) = EMISSION(1,NYE) - FFER
END IF
END IF
C — 1015 gC/yr -> ppm/yr
EMCO2(IT) = EMCO2(IT)/2.123
C
C — End emissions -----
C
C — Calculate SCO2(IT) (=dpCO2,a - dpCO2,s)
SS1 = 0.5*EMCO2(1)/(A*K)
IF (IT .EQ. 1) THEN

```

```

SS2 = 0.5*EMCO2(IT)/(A*K)
SUMS = 0.0
ELSE
SS2 = 0.5*EMCO2(IT)/(A*K) - YCO2(IT-1)/(DT*A*K)
SUMS = SUMS + EMCO2(IT-1)/(A*K)-SCO2(IT-1)
END IF
SCO2(IT) = CC1*(SUMS + SS1 + SS2)
C
C — Calculate ZCO2(IT) (=dSCO2, change in total inorganic carbon in the
mixed layer)
C
SUMZ = 0.0
DO I=1,IT-1
ITX = IT-I
SUMZ = SUMZ + SCO2(I)*RS(ITX)
END DO
C
C — Since RS(IT=0) = 1, and SCO2(0) = 0
ZCO2(IT) = C*K*DT/H *(SUMZ + 0.5*SCO2(IT))
c
C — Calculate new YCO2(IT) (= dpCO2,s)
c
YCO2(IT) = 1.3021*ZCO2(IT) + 3.7929E-3*(ZCO2(IT)**2) +
. 9.1193E-6*(ZCO2(IT)**3) + 1.488E-8*(ZCO2(IT)**4) +
. 1.2425E-10*(ZCO2(IT)**5)
C
C — Atmospheric concentration of CO2 (ppm)
XCO2(IT) = SCO2(IT) + YCO2(IT) + 278.
C
END DO
C
C — Update CO2 concentrations in the CONC array.
CONC(1,NY) = XCO2(IT)

```

```

c
RETURN
END
C -----
SUBROUTINE SETUP_CO2
C
INCLUDE 'scm.cmn'
C
YCO2(1) = 0.0
SCO2(1) = 0.0
XCO2(1) = 278.
DFNPP(1) = 0.0
SUMS = 0.0
c
C — Tabulate values of response/decay functions which are only a function
of time.
DO NYEAR=NYSTART,NYEND
IYR = NYEAR-NYSTART+1
DO IDT = 1,IDTM
IT = (IYR-1)*IDTM + IDT
TIME = (IYR-1)+FLOAT(IDT-1)/IDTM
C — Pulse respons function for the mixed layer (t<2 yr)
IF (TIME .LE. 2.0) THEN
RS(IT) = 0.12935 + 0.21898*exp(-TIME/0.034569) +
. 0.17003*exp(-TIME/0.26936) +
. 0.24071*exp(-TIME/0.96083) +
. 0.24093*exp(-TIME/4.9792)
C — Pulse respons function for the mixed layer (t>2 yr)
ELSE
RS(IT) = 0.022936 + 0.24278*exp(-TIME/1.2679) +
. 0.13963*exp(-TIME/5.2528) +
. 0.089318*exp(-TIME/18.601) +
. 0.03782*exp(-TIME/68.736) +

```



```
. 0.035549*exp(-TIME/232.3)
END IF
C — Biotic decay function
RB(IT) = 0.70211*EXP(-0.35*TIME) +
. 13.4141E-3*exp(-TIME/20.) -
. 0.71846*exp(-55*TIME/120.) +
. 2.9323E-3*exp(-TIME/100.)
END DO
END DO
RETURN
END
```

This is CICERO

CICERO was established by the Norwegian government in April 1990 as a non-profit organization associated with the University of Oslo.

The research concentrates on:

- International negotiations on climate agreements. The themes of the negotiations are distribution of costs and benefits, information and institutions.
- Global climate and regional environment effects in developing and industrialized countries. Integrated assessments include sustainable energy use and production, and optimal environmental and resource management.
- Indirect effects of emissions and feedback mechanisms in the climate system as a result of chemical processes in the atmosphere.

Contact details:

CICERO
P.O. Box. 1129 Blindern
N-0317 OSLO
NORWAY

Telephone: +47 22 85 87 50
Fax: +47 22 85 87 51
Web: www.cicero.uio.no
E-mail: admin@cicero.uio.no

